STOP GROWING MUSHROOMS PART III

Tom Knight
Hewlett Packard

In my paper presented at the scrug conference last year I stated that I felt it was all to easy to create problems in your operations department by not keeping your operators informed. Stop growing mushrooms, "keep them in the dark and feed them a lot of *fertilizer".

Thanks to my friend Marc Covitt, we have updated that phrase to "keep them in the dark and feed a lot of *fertilizer and if they grow or prosper, then cut them off at the knees".

These two phrases are some what the same as saying "what they don't know will not hurt them". So again I will try to provide some helpful information in the operations area.

We will spend some time with:

Operations Guidelines

Run Documentation

UDC's and other tricks

Hints on how to keep your operators happy

* not quite the word normally used.

Operations Guidelines

This is an area that most shops really seem to be lacking in. What you'll find is a total lack of guidelines or very poor guidelines.

You can't expect your users to work well with you if you don't provide them with information on what you expect of them.

Example 1:

```
:FILE T;DEV=TAPE
:STORE PINCO@.DATA.PROD;*T;SHOW


:TELLOP ;
:TELLOP ; ***********************************
:TELLOP ; * MOUNT 2400' SCRATCH TAPE.       *
:TELLOP ; * LABEL IN TAPELIB AS PIN0001T.    *
:TELLOP ; ***********************************
:TELLOP ;
:FILE PIN0001T;DEV=TAPEOUT;DEN=1600
:RUN TS0065P.UTIL.SYS;INFO="PINCO@.DATA.PROD;*PIN0001T; &
:      SHOW;CONSOLE;CHECK=10"
```

Note

TS0065P is a utility program written by Joe Fleming at Hewlett Packard San Diego Division which is safe to run (runs in user mode). After checking for the extended keywords, the STORE COMMAND is passed to MPE via the COMMAND INTRINSIC. This Program will be on the latest SWAP Tape, with two versions which will allow it to run on either Ciper or Q-Mit.

Both methods work, and will be accepted by MPE but, you can bet the second method will work far better and create less chances for something to go wrong.

Example 2:

```
:FILE PRT;DEV=LP;CCTL

:COMMENT ; ********************************************
:COMMENT ; * REPORT PIN2202R CREATES OVER 12000 SECTORS *
:COMMENT ; * OF SPOOL FILES. PRINT AT YOUR CONVIENCE ON *
:COMMENT ; * ON STANDARD 1-PART PAPER   8 LPI.         *
:COMMENT ; ********************************************
:COMMENT ;
:FILE PIN2202R;DEV=LP,7,1;CCTL
```

In both examples, it's plain to see that you can force your operators into their mind reading act or you can inform them of what's supposed to happen.

Tell

Tell them what you expect.
Tell them how to do it.
Tell them if they did it right.

Sample JCL Implementation Plan

```
  I.  Prepare JCL as per New Standards
      A.  USE file  (see Use file documentation)
      B.  TDP or other means

 II.  Submit JCL "Package"  to operations for review
      A.  hardcopy JCL listing
      B.  move file              (if needed)
      C.  run documentation      (as needed - per operations request)
      D.  large jobs             (require 5 day minimum in advance)
      E.  small jobs             (require 2 day minumum in advance)

III.  Acceptance/Rejection
      A.  JCL listing returned to analyst
           1.  acceptance
           2.  rejection and reasons why
           3.  analyst makes corrections and resubmits
                a.  gets higher priority

 IV.  Exceptions
      A.  "HOT" production jobs
           1.  accepted immediately after change
           2.  PR generated to bring JCL to standards
```

V. Key JCL for Renovation
   A. based on top percentage of problem reports (PR's)
   B. 1 JCL/group/week
   C. all JCL changed as problems occur
   D. all JCL related to any MR

All stages are required except those specifically noted as optional.

Stages 1-7 and 21-27 are Job related.
Stages 8-20 are step related.

Stage	Command	Explanation
1	!JOB	Conform to naming standards must be the same as the JCL file
2	!COMMENT	1-3 lines describing the job.
3	!COMMENT	Frequency daily/weekly/monthly
4	!COMMENT	Run time 10 minutes, 2 hours, etc.
5	!COMMENT	Author and date written
6	!COMMENT	JCL revision history: name, date, brief description
7	!COMMENT	All input files and data set/base names and their source
8	!TELLOP !TELLOP;********** !TELLOP;* JOB= jobname step=step-no phase=program * !TELLOP;* description of the step and/or operator action * !TELLOP;********** !TELLOP	
9	!SHOWTIME	
10	!SHOWJOB	
11	!RESET@	
12	!SETJCW	Optional
13	!PURGE	Purge a file before attempting a !BUILD
14	!COMMENT	A brief description of why the following !BUILD statements are not blocked the same as the BLOCK program recommends
15	!BUILD	For all output disc files !BUILD FILE XX;DISC=NN,32,16;ETC where NN = 2 times the antipated maximum volume of records. If this is a temporary step related work file, then use ";TEMP"
16	!FILE	Required for all files of any file type for each step. For output disc files use: !FILE FILE-a=FILE-x,OLD or OLDTEMP to reference the previous !BUILD
17	!COMMENT	For restart instructions

```
18              !CONTINUE         Optional
19              !RUN Program
20              !IF               Optional

21              !LISTF listf,2    For all output disc files

22              !PURGE            For input and output files not
                                  required for another job

23              !COMMENT          Used if a file is not purged,
                                  give reason why it was purged

24              !RUN SCRUNCH      For all output disc files not purged

25              !COMMENT          Used if a file was not SCRUNCHed

26              !TELLOP;
                !TELLOP; ****************************************
                !TELLOP; *     ///// jobname  IS  E  O  J /////     *
                !TELLOP; ****************************************
                !TELLOP;

27              !EOJ
```

Example 3:

A sample JCL using this standard

```
!JOB ICA2002J,JOBS.PROD,ICAPPS
!COMMENT ****************************************************
!COMMENT         JOB DESCRIPTION
!COMMENT ICAPPS DAILY BATCH REPORTS, INCLUDING WORKORDER DOC.
!COMMENT
!COMMENT RUN FREQUENCY:  DAILY
!COMMENT JOB TAKES APPROX.:  1:30 HRS.
!COMMENT WRITTEN BY: DORENE MATNEY    02-18-83
!COMMENT ****************************************************
!COMMENT        JCL REVISION HISTORY      (NAME, DATE, DESCRIPTIO
!COMMENT
!COMMENT   02/23/83   DORENE MATNEY   (M968)
!COMMENT      -ADDED STEPS 062, 064, 066, 068, 069 TO PRINT
!COMMENT       MATERIAL LISTS AS PART OF WORKORDER PACKET
!COMMENT      -ADD FILE MRF0156D FROM WOF0230P AS MATERIAL
!COMMENT       LIST REQUEST FILE
!COMMENT ****************************************************
!COMMENT     INPUT FILES    USED BY THIS JOB AND THEIR SOURCE
!COMMENT MRF0152C (CONTROL CARD USED BY MRF1100P--FROM SADIE)
!COMMENT MRF0700P, MRF1100P, MRF0940P (FROM MRF0001J--SADIE)
!COMMENT WOF2300D (ICA2001J)
!COMMENT WOF0014D
!COMMENT XEQST503
!COMMENT ICA3020D (ICA2001J)
!COMMENT PINCO
!COMMENT MRFBAS
!COMMENT ICAPPS
!COMMENT
!TELLOP *=========================================================
!TELLOP *   JOB=ICA2002J  STEP=010  PROGRAM=ICA2010P.PROG
!TELLOP * STRIP CURRENT CONTROLLER ASSIGNMENTS
!TELLOP *=========================================================
!SHOWTIME
!SHOWJOB
!RESET @
!COMMENT
!COMMENT
!COMMENT
```

```
!   FILE PINCO.PROD.DATABASE=PINCO.DATA.PROD
!COMMENT ************************************************************
!COMMENT *   RESTART INSTRUCTIONS:
!COMMENT * RESTART AT THIS STEP
!COMMENT ************************************************************
!   RUN ICA2010P.PROG;LIB=G
!COMMENT
!TELLOP *=============================================================
!TELLOP *   JOB=ICA2002J  STEP=020  PROGRAM=FCOPY.PUB.SYS
!TELLOP * LOADS PINCO GENERATED RUNS FROM TAPE TO DISK
!TELLOP * ---------------------------------------------
!TELLOP * MOUNT TAPE "PIN4603T" CREATED IN PINCO TONIGHT
!TELLOP *              (1600 BPI)
!TELLOP *=============================================================
!SHOWTIME
!SHOWJOB
!RESET @
!   PURGE PIN4603D
!COMMENT
!   BUILD PIN4603D;REC=-80,144,F,ASCII;DISC=1584,12,6
!   FILE PIN4603D,OLD
!   FILE PIN4603T;DEV=TAPEIN;REC=-80,10,F,ASCII
!COMMENT ************************************************************
!COMMENT *   RESTART INSTRUCTIONS:
!COMMENT * RESTART AT THIS STEP
!COMMENT ************************************************************
!   RUN FCOPY.PUB.SYS
FROM=*PIN4603T;TO=*PIN4603D;SUBSET
EXIT
!COMMENT
!TELLOP *=============================================================
!TELLOP *   JOB=ICA2002J  STEP=030  PROGRAM=SORT.PUB.SYS
!TELLOP * SORT TRIGGERS FOR WORKORDER PACKET
!TELLOP *=============================================================
!SHOWTIME
!SHOWJOB
!RESET @
!   PURGE TRIGSORT
!COMMENT
!COMMENT
!   FILE INPUT=WOF2300D,OLD
!   FILE OUTPUT=TRIGSORT
!COMMENT ************************************************************
!COMMENT *   RESTART INSTRUCTIONS:
!COMMENT * RESTART AT THIS STEP
!COMMENT ************************************************************
!   RUN SORT.PUB.SYS
KEY 1,12
END
!   PURGE WOF2300D
!   RENAME TRIGSORT,WOF2300D
!COMMENT
!TELLOP *=============================================================
!TELLOP *   JOB=ICA2002J  STEP=040  PROGRAM=WOF0230P.PROG
!TELLOP * FORMATS WORKORDERS FOR PRINTING
!TELLOP *=============================================================
!SHOWTIME
!SHOWJOB
!RESET @
!   PURGE WOF2301D
!   PURGE MRF0156D
!COMMENT
!   BUILD WOF2301D;REC=-94,128,F,ASCII;DISC=48000,32,16
!   BUILD MRF0156D;REC=-80,160,F,ASCII;DISC=3040,20,10
!   FILE WOF2301D,OLD
!   FILE MRF0156D,OLD
!   FILE WOF2300D,OLD
```

```
!    FILE PIN4603D,OLD
!    FILE PINCO.DATA=PINCO.DATA.PROD
!    FILE MRFBAS.DATA=MRFBAS.DATA.PROD
!COMMENT ********************************************************
!COMMENT *    RESTART INSTRUCTIONS:
!COMMENT * RESTART AT THIS STEP
!COMMENT ********************************************************
!    RUN WOF0230P.PROG
!COMMENT
!TELLOP *=======================================================
!TELLOP *    JOB=ICA2002J  STEP=050  PROGRAM=WOF0240P.PROG
!TELLOP * PRODUCES HARDCOPY WORKORDERS
!TELLOP *=======================================================
!SHOWTIME
!SHOWJOB
!RESET @
!    PURGE WOF0017D
!COMMENT
!    BUILD WOF0017D;REC=-82,128,F,ASCII;DISC=5120,21,11
!    FILE WOF0017D,OLD
!    FILE WOF2301D,OLD
!    FILE WOF0014D,OLD
!    FILE WORKORDS;DEV=FORMS,6,1;CCTL;FORMS=MOUNT WORKORDER FORMS
!COMMENT ********************************************************
!COMMENT *    RESTART INSTRUCTIONS:
!COMMENT * RESTART  AT THIS STEP
!COMMENT ********************************************************
!    RUN WOF0240P.PROG



!    PURGE ICA0250D
!    PURGE ICA0251D
!    PURGE ICA2401D
!    PURGE ICA2401W
!    PURGE ICA3020D
!    PURGE ICA3090D
!    PURGE ICA3091W
!    PURGE ICA3092W
!    PURGE ICA3093W
!    PURGE ICA3094W
!    PURGE MRF0156D
!    PURGE MRF0700D
!    PURGE MRF0700W
!    PURGE MRF0701W
!    PURGE MRF0701D
!    PURGE PIN4603D
!    PURGE WOF0017D
!    PURGE WOF2301D
!    PURGE WOF0250D
!    PURGE WOF2300D
!COMMENT
!COMMENT
!COMMENT
!TELLOP *=============================================*
!TELLOP *     /////    END OF ICA2002J     /////
!TELLOP *=============================================*
!SHOWTIME
!EOJ
```

Run Documentation

Now here is an area that programmers have become very famous for. If given the chance most programmers would provide little or no documentation with new programs. I feel I should make it clear that most programmers because of time constraints and massive work loads would like to avoid spending time creating run documentation, but the job's not done until the paperwork is complete. Although time consuming, this can make life better for the programmers. With good JCL and QUALITY DOCUMENTATION can avoid some of those late night or weekend (vacation) calls from your bewildered operators.

COMPUTER DOCUMENTATION

PAGE 1 of 9

JOB NAME: ICA2002J PROGRAMMER: DORENE MATNEY

RUN TIME: 1 1/2 HOURS LOGON: OPS.PROD

FREQUENCY: DAILY COMMENTS: LET ONLINE USERS ON

SYSTEM: GEORGE

STEP: 010
PHASE NAME: ICA2010P PINCO
CONTROL
DESCRIPTION: STRIP MASTER

CONTROLLER ASSIGNMENTS

OUTPUT DISTRIBUTION: ICA2010P

TAPES: N/A

REPORTS: N/A WOF0014D

FICHE: N/A

STEP: 020
PHASE NAME: FCOPY PIN4603T

DESCRIPTION: LOAD PINCO

RUN DATA TO DISC

OUTPUT DISTRIBUTION: FCOPY

TAPES: N/A

REPORTS: N/A PIN4603D

FICHE: N/A

UDC's

OPSUDC1.OPS.PROD

STARTUP	USER
CMD	USER
INIT	USER
AJ	USER
JAM	USER

```
    W                   USER
    RE                  USER
    SO                  USER
    SOR                 USER
    AF                  USER
    AF19                USER
    AF6                 USER
    ASF                 USER
    R                   USER
    WEL                 USER
    HOFF                USER
    HON                 USER
    LIM                 USER
    DEL                 USER
    B                   USER
    0                   USER
    SUS                 USER
    UNL                 USER
    SYSMON              USER
    TERM                USER
    SL                  USER
    SLOG                USER

SYSTEMU.PUB.SYS
    A                   SYSTEM
    ASK                 SYSTEM
    B                   SYSTEM
    BANNER              SYSTEM
    BANNERCOB           SYSTEM
    BANNERFTN           SYSTEM
    BANNERSPL           SYSTEM
    BLD                 SYSTEM
    BLOCK               SYSTEM
    CMD                 SYSTEM
    CREATOR             SYSTEM
    DATES               SYSTEM
    DBSHOW              SYSTEM
    DBUTIL              SYSTEM
    DECOMP              SYSTEM
    DBWHO               SYSTEM
    DIR2                SYSTEM
    EDIT2               SYSTEM
    END                 SYSTEM
    FILES               SYSTEM
    FINDFILE            SYSTEM
    FORMSPEC            SYSTEM
    GRAPH               SYSTEM
    HARDCOPY            SYSTEM
    HPDRAW              SYSTEM
    HPSLATE             SYSTEM
    HPWORD              SYSTEM
    LISTFILE            SYSTEM
    OPT                 SYSTEM
    OPTICALC            SYSTEM
    PURG                SYSTEM
    Q                   SYSTEM
    QUERY               SYSTEM
    RP                  SYSTEM
    RU                  SYSTEM
    SEG                 SYSTEM
    SJ                  SYSTEM
    SJS                 SYSTEM
    SP                  SYSTEM
    SPOOK               SYSTEM
    ST                  SYSTEM
    STREAMER            SYSTEM
    TDP                 SYSTEM
```

TOOLSET SYSTEM

```
STARTUP
OPTION LOGON,NOBREAK
RUN ALLOWME.UTIL.SYS;PARM=1
CONTINUE
FILE STORE;DEV=TAPE
FILE RESTORE;DEV=TAPE
FILE SYSLIST;DEV=LP,7,1
RUN LISTEQ2.PUB.SYS
****
INIT
OPTION LIST
LIM
JOBFENCE 5
OUTFENCE 7
OUTFENCE 8;LDEV=19
STREAMS 10
COMMENT SYSTEM IS UP AND RUNNING
****
AJ PARM
ABORTJOB  !PARM
****
JAM PARM
SUSPENDSPOOL !PARM
OPTION LIST
CONTINUE
RESUMESPOOL !PARM
****
W PARM=0
SETJCW WARNING = !PARM
IF WARNING = 3 THEN
    WARN @;MRF STARTING -
           PLEASE EXIT THE MRF DATABASE
ELSE
IF WARNING = 5 THEN
    WARN @;NIGHTLY PRODUCTION STARTING
           PLEASE LOGOFF
    ENDIF
ENDIF
****
RE
RECALL
****
SO
SHOWOUT ACTIVE
CONTINUE
SHOWOUT READY
****
SOR PARM
SHOWOUT JOB= J!PARM;READY
****
AF PARM
OPTION LIST
ALTSPOOLFILE  0!PARM;PRI=9
****
AF19 PARM
OPTION LIST
ALTSPOOLFILE  0!PARM;DEV=19
****
AF6 PARM
OPTION LIST
ALTSPOOLFILE  0!PARM;DEV=6
****
ASF PARM PARM1 PARM2 PARM3=1
```

```
OPTION LIST
ALTSPOOLFILE  0!PARM;DEV=!PARM1;
              PRI=!PARM2;COPIES=!PARM3
****
R PARM PARM1
REPLY !PARM,!PARM1
****
WEL
WELCOME
****
HOFF PARM
OPTION LIST
HEADOFF !PARM
****
HON !PARM
OPTION LIST
HEADON !PARM
****
LIM PARM=2 PARM1=64
OPTION LIST
LIMIT !PARM,!PARM1
****
DEL PARM
OPTION LIST
DELETESPOOLFILE  !PARM
****
B PARM
BREAKJOB  !PARM
****
O PARM
OPTION LIST
OUTFENCE !PARM
****
SUS PARM
SUSPENDSPOOL !PARM
****
TERM PARM
FILE TERM,NEW;REC=-132,,F,ASCII;DEV=!PA
RUN POW0001P.JCL.TEST
****
SL
SHOWLOG
****
SLOG
SHOWLOGSTATUS
****


STARTUP
OPTION LIST,LOGON,NOBREAK
CONTINUE
LIMIT 0,0
CONTINUE
OUTFENCE 12
CONTINUE
JOBFENCE 14
CONTINUE
STREAMS 10
CONTINUE
TUNE 1000;CQ=152,200,0,300;DQ=202,238,1
CONTINUE
FILE STMFILE=SYS0200J.JCL.PROD
CONTINUE
COMMENT RUN TS0020P.UTIL.SYS;PARM=1
CONTINUE
FILE STRMFILE=SYS0001J.SUPPORT
CONTINUE
```

```
RUN STREAMX.PUB.VESOFT;PARM=1
CONTINUE
DSCONTROL HAL;OPEN,NOW
CONTINUE
DSCONTROL KERMIT;OPEN,NOW
CONTINUE
DSCONTROL GEORGE;OPEN,NOW
CONTINUE
DSCONTROL SADIE;OPEN,NOW
CONTINUE
BYE
***

!JOB SYS0001J,JOBS.PROD,EDP
!COMMENT ******************************
!COMMENT *                JOB DESCRIPTIO
!COMMENT * This job stream is part of t
!COMMENT * procedure. A log-on UDC for
!COMMENT * essential commands, then str
!COMMENT * allocates space for system u
!COMMENT *
!COMMENT * RUN FREQUENCY:  With every s
!COMMENT * JOB TAKES APPROX.:  2 minute
!COMMENT * WRITTEN BY: Jeannine Smith
!COMMENT ******************************
!COMMENT
!COMMENT ******************************
!COMMENT *   JCL REVISION HISTORY
!COMMENT *
!COMMENT *
!COMMENT *
!COMMENT ******************************
!COMMENT

!TELLOP ******************************
!TELLOP *   JOB=SYS0001J
!TELLOP *   ALLOCATEs system utilities
!TELLOP ******************************
!SHOWTIME
!SHOWJOB
!RESET @
!ALLOCATE EDITOR.PUB.SYS
!CONTINUE
!ALLOCATE QUERY.PUB.SYS
!CONTINUE
!ALLOCATE LISTDIR2.PUB.SYS
!CONTINUE
!ALLOCATE TS0020P.UTIL.SYS
!CONTINUE
!ALLOCATE LOGON.UTIL.SYS
!CONTINUE
!ALLOCATE INTRFACE.UTIL.SYS
!CONTINUE
!ALLOCATE LISTER.UTIL.SYS
!CONTINUE
!ALLOCATE TDP.PUB.SYS
!CONTINUE
!ALLOCATE DSCOPY.PUB.SYS
!CONTINUE
!ALLOCATE SCRIBE.PUB.SYS
!CONTINUE
!ALLOCATE FCOPY.PUB.SYS
!CONTINUE
!ALLOCATE TSJ0600P.UTIL.SYS
!CONTINUE
!ALLOCATE OPT.PUB.SYS
!CONTINUE
```

```
!ALLOCATE TS0750P.UTIL.SYS
!CONTINUE
!ALLOCATE USER.PUB.SECURITY
!CONTINUE
!ALLOCATE STORE.PUB.SYS
!CONTINUE
!COMMENT
!TELLOP ********************************
!TELLOP *       END OF SYS0001J
!TELLOP ********************************
!SHOWTIME
!EOJ
```

HINTS ON KEEPING YOUR OPERATORS HAPPY!

At Hewlett Packard we use a system called Management by Objective If management has established the plans, objectives and goals of the company and department, then the individual can set his or her goals with the best interests of all concerned. Point, if management can efficiently handle:

Planning

Organizing

Controlling

Leading

Then the employees can set their goals to emphasize what should be done rather than how it is to be done, leaving room for creative solutions.

Since most EDP departments tend to run on a 24 hour 7 - day schedule, some very real problems can and most often occur. It's not unusual for an operator(s) to be be working with little or no supervision during the night shift and weekends. This makes it very important that supervision provide a constant flow of information to and from the operators. More often than not I have found that once an employee ends up on night shift, they tend to hear from supervision only when they do something wrong. When an employee hears only negative thoughts, then they tend to go astray.

One of the things that we have found at San Diego Division to help keep operators happy is to provide special projects for them to perform.

A project can range from something very simply like keeping an inventory of printer ribbons and reordering when needed, to writing complex programs. This requires that supervision spend just a little time with the operators on an ongoing basis to learn there wants, desires and needs. This time spent can provide a bonus for both the employee and the company, by providing a vehicle of upward learning for the employee and better performance which helps the company.